

**R. Brigola, TH Nürnberg Georg Simon Ohm, 2014**

**Mathematica - Notebooks als Bonusmaterial zum Lehrbuch**

**[1] Rolf Brigola Fourier-Analysis und Distributionen,  
Eine Einführung mit Anwendungen,  
edition swk, Hamburg 2013**

**Teil 5 Entwurf von analogen Tschebyscheff-Tiefpassfiltern  
(Chebyshev Lowpass Filter) vom Typ I mit *Mathematica***

Referenzen zu Kapiteln, Abschnitten, Seitenzahlen beziehen sich im Folgenden meist auf das genannte Lehrbuch des Autors. Einige wenige weitere Referenzen findet man am Ende des Notebooks.

Die URL aller meiner *Mathematica*-Notebooks zur Fourier-Analysis ist:  
<http://www.stiftung-swk.de/mathematica/>

In diesem Notebook wird gezeigt, wie man ein analoges Tschebyscheff-Tiefpassfilter zu einem vorgegebenen Dämpfungsplan entwirft. *Dabei folgen wir der Aufgabenstellung in [1], Aufgabe A22 zu Kap. 5, d.h. wir beschränken uns auf sogenannte Tschebyscheff-Tiefpässe vom Typ I.*

Aus den Koeffizienten des berechneten Frequenzgangs kann ein Ingenieur der Elektrotechnik das Filter mit Widerständen, Kapazitäten und Operationsverstärkern z.B. mit der Sallen-Key-Filtertopologie leicht als Schaltung realisieren (vgl. etwa <http://de.wikipedia.org/wiki/Sallen-Key-Filter> ). Aus dem analogen Tiefpassfilter kann man *durch bilineare Transformation oder andere Verfahren ein diskretes Filter* mit einem entsprechenden Frequenzgang gewinnen. Ich gehe darauf in einem anderen Notebook, das den Inhalten von [1], Kapitel 10 und damit wichtigen **Grundlagen der Linearen Systemtheorie** gewidmet sein wird, näher ein. Sie finden es unter der gleichen URL wie dieses Notebook.

Tschebyscheff-Tiefpassfilter bieten nach der Extremaleigenschaft, die für Tschebyscheff-Polynome im meinem vorangehenden Notebook ([http://www.stiftung-swk.de/mathematica/Chebyshev\\_Polynome-Teil1.nb](http://www.stiftung-swk.de/mathematica/Chebyshev_Polynome-Teil1.nb) ) genannt war, einen optimal schnell gedämpften Amplitudengang außerhalb des Passbands im Vergleich zu Filtern mit anderen rationalen Frequenzgängen. Der Preis ist eine "Welligkeit im Passband", das heißt gewisse Verzerrungen, die dadurch zustande kommen, dass das Tschebyscheff-Polynom vom Grad  $n$  in  $[-1, 1]$  (d.h. für  $|\omega/\omega_g| \leq 1$ , also  $\omega$  im Passband) ja  $n$  Nullstellen besitzt, wegen Stetigkeit also dort "wellig sein muss".

Wir beginnen damit, das Filter zunächst eigenständig zu berechnen und sehen am Ende, was die hierfür schon vorhandenen *Mathematica*-Routinen im Vergleich ergeben.

**Anmerkung des Autors:** Wie schon in vorangehenden Notebooks bemerkt, bin ich kein Experte für die schier unerschöpflichen Möglichkeiten, die ein wirklich geübter, vertiefter Umgang mit *Mathematica* bietet. Ich habe mich daher im Wesentlichen bemüht, den Stoff mit diesem Angebot zu veranschaulichen und zu zeigen, wie man mit (zum Teil wahrscheinlich oft zu umständlichen, dafür aber auch für *Mathematica*-Anfänger wie mich selbst transparenten) Anweisungen die behandelten Inhalte erschließen kann.

## Tschebyscheff – Tiefpassfilter vom Typ I

(vgl. [1], Aufgabe A22 von Kap. 5, S. 102)

### 1. Frequenzgang zu gegebenem Dämpfungsplan

Das Quadrat des Amplitudengangs zum Frequenzgang  $\hat{h}(\omega)$  mit der Kreisfrequenz  $\omega$  für ein Tschebyscheff-Tiefpassfilter vom Typ I der Ordnung  $n$  ist definiert durch

$$(1) \quad |\hat{h}(\omega)|^2 = \frac{1}{(1+c^2 T[n,\omega/\omega_g]^2)}$$

Dabei ist  $\omega_g$  die gewünschte *Grenzkreisfrequenz* für das sog. Passband (cutoff angular frequency) und  $c$  ein zunächst freier Parameter.

$T[n,x]$  ist das Tschebyscheff-Polynom erster Art von der Ordnung  $n$ .

Die Ordnung  $n$  und der Parameter  $c$  sind aus dem Dämpfungsplan, den das Filter erfüllen soll, zu berechnen. Damit folgen dann die Filterkoeffizienten. Natürlich soll die Ordnung möglichst niedrig sein, da Bauteile zur Realisierung ja etwas kosten.

**Wir geben als Beispiel folgenden Dämpfungsplan vor:**

Mit  $A=20 \log_{10}\left(1/|\hat{h}(\omega)|\right)$  als Dämpfungsmaß in dB möge für das gesuchte Filter gelten:

$A_{\max} = 0.2$  dB maximale Dämpfung bei der Grenzfrequenz  $f_g = 1.0$  kHz ,

*d.h.* an der Passbandecke mit  $\omega_g = 2000 \pi$  rad/s,

$A_{\min} = 40$  dB minimale Dämpfung an der Stopbandecke  $f_s = 2.5$  kHz.

Außerhalb von  $[-1,1]$  ist  $T[n,x]=\cosh(n \operatorname{arccosh}(x))$ . Damit wird nun das Filter wie folgt berechnet:

Für die gesuchten Größen  $c$  und  $n \in \mathbb{N}$  folgt mit kurzer Rechnung aus den Eingangsdaten  $f_s, f_g, A_{\max}, A_{\min}$  nach unserem Dämpfungsplan:

```
In[345]:= ClearAll["Global`*"]; Remove["Global`*"]; ?Global`*
fg := 1000; fs := 2500; Amax := 0.2; Amin := 40;
c = Sqrt[10^(Amax/10) - 1]
n = Ceiling[N[ArcCosh[Sqrt[(10^(Amin/10) - 1)]]/c] / ArcCosh[fs/fg]]]
(* Filterordnung *)
```

... **Information:** No symbol matching Global`\* found.

Out[347]= 0.217091

Out[348]= 5

Weil das Argument von Ceiling i.A. nicht ganzzahlig ist und wir eine ganze Zahl als Filterordnung verwenden, ergibt sich durch die ggf. nötige Aufrundung eine kleine Abweichung von den Spezifikationsdaten, die aber in der Regel eine Verbesserung der Dämpfungseigenschaften des Filters sein wird.

Für den **Frequenzgang des Filters** verwendet man die *Polstellen*  $z_k$ ,  $k = 1, \dots, n$  mit *negativen Realteilen* der Funktion  $Q(z)=H(z)H(-z)=1/(1+c^2 T[n, z/(i\omega_g)]^2)$ ,  $\omega_g$  die Grenzkreisfrequenz. *Es gibt genau n Pole mit negativem Realteil*. Damit erhalten wir ein stabiles lineares Filter der Ordnung  $n$ . Der resultierende Frequenzgang als Funktion der Kreisfrequenz  $\omega$  ist dann mit  $\hat{h}(\omega)=H(i\omega)$

$$\hat{h}(\omega)=K \prod_{k=1}^n \frac{z_k}{z_k-i\omega}.$$

Wir wählen die Gleichspannungsverstärkung  $K$  (dc-gain) gleich 1 bei ungerader Filterordnung  $n$ , andernfalls  $K=1/\sqrt{1+c^2}$ .

Die Funktion  $H(s)$ ,  $s \in \mathbb{C}$ , als der Faktor von  $Q$  mit Polen, die negative Realteile haben, heißt **Übertragungsfunktion des Filters**.

## 1.1 Berechnung der Polstellen für die Übertragungsfunktion, Übertragungsfunktion und Frequenzgang

```
In[349]:= sol = NSolve[1 + c^2 ChebyshevT[n, z / (1 * 2 * Pi * fg)]^2 == 0 && Re[z] < -10^(-5), z]
(* gesuchte Pole *)
z = Table[sol[[k, 1, 2]], {k, 1, n}] (* hier nochmal als Liste *)
```

Out[349]= {{z -> -2899.13}, {z -> -2345.44 - 4067.34 i},  
{z -> -2345.44 + 4067.34 i}, {z -> -895.88 - 6581.1 i}, {z -> -895.88 + 6581.1 i}}

Out[350]= {-2899.13, -2345.44 - 4067.34 i,  
-2345.44 + 4067.34 i, -895.88 - 6581.1 i, -895.88 + 6581.1 i}

Wir halten fest, dass bei ungerader Ordnung ein Pol reell ist, die anderen treten zusammen mit den konjugiert komplexen Polstellen auf. Bei gerader Filterordnung kommt kein reeller Pol vor. Die resultierende rationale Übertragungsfunktion hat

also im Nenner ein Polynom mit reellen Koeffizienten, was für eine Realisierung ja auch wichtig ist.

Wir berechnen die Pole nochmals analytisch und verändern wir die Darstellung etwas, indem wir die Übertragungsfunktion im Nenner so darstellen, dass die Linearfaktoren zu konjugiert komplexen Polen zu einem quadratischen Faktor mit reellen Koeffizienten zusammengefasst werden. Das Filter kann man dann als Serienschaltung von sog. "Biquads", dh. Filtern 2-ter Ordnung mit diesen schon gegebenen quadratischen Faktoren und im Fall ungerader Filterordnung mit einem zusätzlichen Glied erster Ordnung realisieren.

Nachfolgend von den gesuchten Polen mit negativem Realteil nur diejenigen mit Imaginärteil  $\geq 0$  in der Table Z (man vgl. oben Table z). Damit wird H so angegeben, dass wir die "Biquads" im Nenner der Übertragungsfunktion sofort sehen. Aus den Koeffizienten kann eine realisierende Anlogschaltung direkt abgeleitet werden.

```
In[351]:= If[EvenQ[n], K := 1/Sqrt[1 + c^2], K := 1] (* dc-gain *)
Y := -1/n Log[1/c + Sqrt[1/c^2 + 1]];
If[EvenQ[n], m := n/2, m := Floor[n/2] + 1]
(* m = Anzahl Pole mit Imaginärteil >= 0 *)
X := Table[(1 + 2 (k - 1)) pi / (2 n), {k, 1, m}]
Z = Table[2 pi fg (Sin[X[[k]]] Sinh[Y] + I Cos[X[[k]]] Cosh[Y]), {k, 1, m}]
(* Pole mit Imaginärteil >= 0 *)
HH[s_] := N[K Product[(Z[[k]] Conjugate[Z[[k]])] /
  (Expand[(s - Z[[k]]) (s - Conjugate[Z[[k]])])], {k, 1, Floor[n/2]}]]
If[EvenQ[n] == False, H[s_] = HH[s] (Z[[m]] / (Z[[m]] - s)), H[s_] = HH[s]]
(* H nun die Übertragungsfunktion,  $\hat{h}(\omega) = H(i\omega)$  der Frequenzgang des Filters *)

Out[355]= {-895.88 + 6581.1 i, -2345.44 + 4067.34 i, -2899.13}

Out[357]= -((2.81927 x 10^18 + 0. i) / ((-2899.13 - s) ((4.41135 x 10^7 + 0. i) + (1791.76 + 0. i) s + s^2)
  ((2.20444 x 10^7 + 0. i) + (4690.89 + 0. i) s + s^2)))
```

Hier sehen wir also die Übertragungsfunktion  $H[s]$  zum **Frequenzgang  $\hat{h}(\omega) = H(i\omega)$** .

Schlägt man das Vorzeichen des Quotienten zum ersten Linearfaktor im Nenner, dann sieht man, dass das Nennerpolynom bei positivem "dc-gain" K lauter positive reelle Koeffizienten besitzt. Gleiche Vorzeichen bei diesen Koeffizienten sind bekanntlich eine notwendige Bedingung für die Stabilität des Filters.

**Bemerkung:** Der Frequenzgang eines stabilen kausalen linearen Filters, dessen Übertragungsverhalten "im Zeitbereich" durch eine lineare Differentialgleichung mit konstanten Koeffizienten beschrieben wird, ist die **Fouriertransformierte  $\hat{h}$  der (kausalen) Impulsantwort  $h$  des Filters** (LTI-Systems). Umgekehrt kann man zu einer rationalen Übertragungsfunktion eine solche Differentialgleichung angeben, aus deren Koeffizienten die schaltungstechnische Realisierung abgeleitet werden kann.

Das Nennerpolynom der Übertragungsfunktion ist dann gerade das charakteristische Polynom der Differentialgleichung, ein ggf. vorhandenes Zählerpolynom gehört zu einem linearen Differentialoperator auf der rechten Seite der Differentialgleichung mit den Koeffizienten dieses Zählerpolynoms. Mehr dazu in meinen Notebooks zur Fouriertransformation und zeitinvarianten linearen Systemen (vgl. [1], Kap. 9-10).

Wir testen kurz die Übereinstimmung von  $|H[i \omega]|^2$  mit der rechten Seite in (1) oben, indem wir ein paar Werte der Differenzfunktion prüfen und diese plotten. Bis auf die numerisch bedingten Fehler der Größenordnung  $10^{-14}$  stimmt der berechnete Frequenzgang mit (1) überein.

```
In[358]= H[0]
K
difference[ω_] := H[I ω] H[-I ω] - 1 / (1 + c^2 ChebyshevT[n, ω / (2 π fg)]^2)
difference[2 π 500]
difference[2 π 1000]
difference[2 π 2500]
Plot[difference[2 π f], {f, 0, 2 fs}, PlotRange -> All]
```


```
Out[358]= 1. + 0. i
```

```
Out[359]= 1
```

```
Out[361]= 0. + 0. i
```

```
Out[362]= 4.44089 × 10-16 + 0. i
```

```
Out[363]= -1.69407 × 10-21 + 0. i
```



## 1.2 Grafische Darstellungen des Amplitudengangs, des Phasengangs, der Verzögerung (Delay) und der Gruppenlaufzeit (Group Delay)

Zunächst zum Test ein paar Werte:  $-20\text{Log}[10, \text{Abs}[H[0]]]$  dc-gain,  $-20\text{Log}[10, \text{Abs}[H[2\pi f_g]]]$  Dämpfung bei der Grenzfrequenz,  $-20\text{Log}[10, \text{Abs}[H[2\pi f_s]]]$  Dämpfung an der Stopbanddecke, alle in dB gemessen.

Danach der Amplitudengang für Frequenzen  $f \in [0, 2 f_s]$  in Hz.

Dazu in rot als Geradenstück der Wert, der an der Grenzfrequenz  $f_g$  angenommen wird, und der Wert an der Stopbandecke bei  $f_s$ .

Die schon angesprochenen leichten Verzerrungen im Passband als Preis für das gute Dämpfungsverhalten außerhalb des Passbands ist deutlich zu sehen.


```
In[365]:= -20 Log[10, Abs[H[0]]]
-20 Log[10, Abs[H[2 π i fg]]] (* Dämpfung in dB an der Passbandecke *)
-20 Log[10, Abs[H[2 π i fs]]] (* und an der Stopbandecke *)
```

Out[365]= 0.

Out[366]= 0.2

Out[367]= 48.7575

```
In[368]:= p1 := Plot[Abs[H[i 2 π f]], {f, 0, 2 fs},
  PlotStyle → Directive[Blue, Thickness[0.005]], PlotRange → All]
p2 := Plot[Abs[H[i 2 π fg]], {f, 0, fg},
  PlotStyle → Directive[Red, Thickness[0.003]], PlotRange → All]
Show[
  p1,
  p2]
```




In der Ingenieurliteratur wird die Amplitudenverzerrung meist so angegeben, dass die Grenzfrequenz auf Eins normiert wird und die Verzerrung in Dezibel gezeigt wird. Diese Darstellung daher hier wie folgt. Sie zeigt sofort, dass das Filter die gewünschte Spezifikation erfüllt.

```

In[371]:= Plot[20 Log[10, Abs[H[i 2 π s fg]]], {s, 0, 2 fs / fg}, Frame → True,
  FrameStyle → Directive[Blue, FontSize → 12, FontWeight → Plain],
  FrameLabel -> {"One-sided Amplitude Distortion in dB", ""},
  {"Cutoff Frequency normed to be one", ""}}, GridLines → Automatic,
  PlotStyle → {Blue, Thickness[0.005]}, PlotRange → All]

```



Nun der Phasengang, wieder für Frequenzen  $f \in [0, 2 fs]$  in Hz, einmal mit  $\phi_1$  mit einem Phasensprung um  $2\pi$ , danach bei der Darstellung mit  $\phi$  die beiden Zweige zu einer stetigen Funktion zusammengefügt. (Die Funktion  $\phi_1$  (rot) ist mit einem offset von  $+0.2$  zur besseren Sichtbarkeit gezeichnet,  $\phi$  ist blau.) Dabei benutzen wir die Darstellung des Nenners von  $H[i\omega]$  als Produkt von Linearfaktoren der Form  $(i\omega - z[[k]])$ ,  $k=1, \dots, n$ . (Der Zähler hat dann das Argument Null.)

```


In[372]:= phi1[w_] := Assuming[Element[w, Reals], Arg[H[I w]]] (* mit Phasensprung *)
phi[w_] = -Sum[Arg[I w - z[[k]]], {k, 1, n}] (* ohne Phasensprung *)
p3 := Plot[phi[2 π f], {f, 0, 2 fs},
  PlotStyle → Directive[Blue, Thickness[0.005]], PlotRange → All]
p4 := Plot[phi1[2 π f] + 0.2, {f, 0, 2 fs},
  PlotStyle → Directive[Red, Thickness[0.005]], PlotRange → All]
Show[
  p3,
  p4]

```

```

Out[373]= -Arg[(895.88 - 6581.1 i) + i w] - Arg[(895.88 + 6581.1 i) + i w] -
  Arg[(2345.44 - 4067.34 i) + i w] - Arg[(2345.44 + 4067.34 i) + i w] - Arg[2899.13 + i w]

```




Nun die **Verzögerung**  $-\text{phi}[2 \pi f]/(2 \pi f)$  in ms (**phase delay**) für Frequenzen  $f \in [0, 2 \text{ fs}]$  in Hz.

Unterschiedliche Verzögerungen im Frequenzband von Audio-Aufnahmen ab etwa 1 ms empfinden wir bereits als reichlich störend - ein gutes Gehör vorausgesetzt. Im HiFi-Bereich sind daher andere Filtertypen (Butterworth-Filter zum Beispiel, vgl. etwa [1], Kap. 10) besser als Tschebyscheff-Filter wie hier gezeigt und nachfolgend erkennbar.

```

In[377]:= delaysms[f_] := -1000 phi[2 π f] / (2 π f)
p5 = Plot[delaysms[f], {f, 0, 2 fs},
  PlotStyle → Directive[Blue, Thickness[0.005]], PlotRange → All]

```




Und schließlich die **Gruppenlaufzeit**  $-D[\text{phi}]$  in ms (**group delay**) für Frequenzen  $f \in [0, 2 \text{ fs}]$  in Hz. Wir differenzieren phi dabei einfach



mal selbst (Nette Übung: Ableitungen von arctan-Funktionen) und plotten den Verlauf.

```
In[379]:= groupdelay[w_] := -Sum[1 / (Re[z[[k]]] + (w - Im[z[[k]])^2 / Re[z[[k]]]), {k, 1, n}]
p6 = Plot[1000 groupdelay[2 Pi f], {f, 0, 2 fs},
PlotStyle -> Directive[Red, Thickness[0.005]], PlotRange -> All]
```




```
In[381]:= groupdelay2[w_] = Sum[D[ArcTan[(w - Im[z[[k]]) / (-Re[z[[k]])]], w], {k, 1, n}]
p7 = Plot[1000 groupdelay2[2 Pi f], {f, 0, 2 fs},
PlotStyle -> Directive[Red, Thickness[0.005]], PlotRange -> All]
```

Out[381]=

$$\frac{0.00111622}{1 + 1.24595 \times 10^{-6} (-6581.1 + w)^2} + \frac{0.000426359}{1 + 1.81782 \times 10^{-7} (-4067.34 + w)^2} +$$

$$\frac{0.000344931}{1 + 1.18978 \times 10^{-7} w^2} + \frac{0.000426359}{1 + 1.81782 \times 10^{-7} (4067.34 + w)^2} + \frac{0.00111622}{1 + 1.24595 \times 10^{-6} (6581.1 + w)^2}$$



### 1.3 Mathematica Transfer Function Models

Wegen der großen Bedeutung linearer Filter in der gesamten Übertragungstechnik, welche heute viele Lebensbereiche durchdringt, bietet *Mathematica* eine ganze Reihe sog. Transfer Function Models an, mit denen ganz unterschiedliche Filtertypen, und zwar sowohl analoge als auch diskrete Filter mit gewünschten Eigenschaften schnell berechnet werden können. Man siehe hierzu die Webseiten

<https://reference.wolfram.com/mathematica/tutorial/AnalogFilterDesign.html>

<https://reference.wolfram.com/mathematica/tutorial/DigitalFilterDesign.html>

Wir betrachten nur kurz als Beispieltyp unser Tschebyscheff-Tiefpassfilter vom Typ I mit den verwendeten Bedingungen und seine *Mathematica*-Berechnung:

```
In[383]:= tf = Chebyshev1FilterModel[{"Lowpass", {2000 Pi, 5000 Pi}, {0.2, 40}]
```

```
Out[383]= 
$$\left( - \frac{2.81927 \times 10^{18} - 145.547 i}{\left( (895.88 - 6581.1 i) + s \right) \left( (895.88 + 6581.1 i) + s \right) \left( (2345.44 - 4067.34 i) + s \right) \left( (2345.44 + 4067.34 i) + s \right)} \right)$$

```

(Weil bei der Wandlung nach pdf der Ausdruck oben leider abgeschnitten wird, hier zur Kenntnis als "plain text"

```
Subsuperscript[(-(2.81927*10^18 + 0. I)/(((895.88 - 6581.1 I) + s) ((895.88 + 6581.1 I) + s)
((2345.44 - 4067.34 I) + s) ((2345.44 + 4067.34 I) + s) (2899.13 + s))), , T]
```

Zum Vergleich die Pole unserer oben berechneten Übertragungsfunktion

```
In[384]:= z
```

```
Out[384]= {-2899.13, -2345.44 - 4067.34 i,
-2345.44 + 4067.34 i, -895.88 - 6581.1 i, -895.88 + 6581.1 i}
```

Sie erkennen, dass wir genau das gleiche Ergebnis "zu Fuß" erhalten haben und so gelernt haben, was hier in einer *Mathematica*-Routine bereits fertig implementiert ist. Über die zahlreichen Möglichkeiten zur Darstellung weiterer Informationen, die man über das Filter ausgeben kann (BodePlot, Polstellen-Diagramm u.a.m.) --> man folge den oben angegebenen Links zu weiteren *Mathematica*-Seiten und Beispielen.

Auf andere Filtertypen als Tiefpassfilter (Bandpässe, Hochpässe, Notchfilter etc.) gehe ich in nachfolgenden Notebooks näher ein, wenn einige Grundlagen zur Fouriertransformation vorausgeschickt sind. Dann werde ich auch entsprechende Beispiele zum Entwurf diskreter Filter an Beispielen mit *Mathematica* demonstrieren, die etwas die mathematischen Hintergründe der fertig vorliegenden *Mathematica*-Modelle beleuchten sollen.

Weitere empfehlenswerte Referenzen:

- [2] U. Tietze, Ch. Schenk     Halbleiter-Schaltungstechnik  
   Springer, 2002
- [3] H. Wupper, U. Niemeyer     Elektronische Schaltungen 2  
   Springer, 1996
- [4] R. Unbehauen                     Systemtheorie  
   Oldenbourg, 2002

u.v.a.m., die Sie im Literaturverzeichnis am Ende von [1] finden können.